Les sections coniques

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

Ce doument est une révision de celui produit en 2008. L'objectif principal de ce document est d'analyser l'équation générale du deuxième degré en x et y.

$$ax^2 + bxy + cy^2 + dx + ey + f = 0$$

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Introduction

Présentation classique des coniques

Lorsque deux cônes circulaires se joignent en leur sommet pour former un cône double appelé cône circulaire droit à deux nappes et qu'ils sont traversés par un plan, on obtient une courbe appelée
- section conique: dans ce cas, il s'agit d'un cercle, ou d'une ellipse, ou d'une parabole ou d'une hyperbole.
- ou section conique dégénérée: dans ce cas, il s'agit d'un point, ou d'une droite, ou de deux droites sécantes.



Section perpendiculaire à l'axe donnant un *cercle*. En haut, section oblique : *ellipse*. En bas, section parall. à la génératrice : *parabole*. En haut et en bas : section parall. à l'axe donnant deux *hyperboles*.

Les coniques ont été largement étudiées par les Grecs qui en découvrirent les propriétés permettant de les définir en termes de points (les foyers, les sommets) et de droites (droite directrice et droites asymptotiques).

Apollonius est l'auteur du seul traité de la Grèce antique sur les sections coniques qui ait survécu au temps. Il est né à Perga, une ville de l'Asie Mineure, au troisième siècle avant Jésus-Christ. Il fit ses études à Alexandrie avec les successeurs d'Euclide. Il a probablement passé la plus grande partie de sa vie à étudier, à enseigner et à écrire dans la ville d'Alexandrie. Son traité, «Les Coniques» représente l'apogée des mathématiques grecques.

Il est difficile, pour les mathématiciens d'aujourd'hui, de comprendre comment Apollonius a pu découvrir et prouver des centaines de théorèmes difficiles sans utiliser le symbolisme de l'algèbre moderne. Et pourtant, il a réussi à le faire magnifiquement.

Utilisation actuelle des coniques

Il est remarquable que, bien qu'étudiées il y a des milliers d'années, les coniques n'aient rien perdu de leur actualité. Elles sont, en effet, indispensables, que ce soit dans la recherche spatiale ou dans l'étude du comportement atomique. On n'a qu'à penser au fait qu'elles servent à décrire le mouvement des planètes, des comètes, des lunes, des astéroïdes et des satellites qui se déplacent dans l'espace et qui sont soumis à des forces gravitationnelles.

Présentation contemporaine des coniques

Aujourd'hui, les coniques sont présentées comme étant l'illustration graphique d'un lieu géométrique définie par une équation quadratique de la forme

$$A x^2 + B x y + C y^2 + \mathrm{D} x + E y + F = 0 \qquad (1)$$

où les constantes A, B et C ne sont pas toutes égales à 0, autrement dit $A^2 + B^2 + C^2 > 0$.

La représentation graphique de l'équation précédente est une conique ou une conique dégénérée. Si c'est une conique, alors c'est soit

- une parabole si $B^2 - 4AC = 0$;
- une ellipse si $B^2 - 4AC < 0$ ou
- une hyperbole si $B^2 - 4AC > 0$.

Dans le cas où $B = 0$, l'équation (1) est une conique ou une conique dégénérée. Si c'est une conique, alors c'est

- une parabole si $AC = 0$, c'est-à-dire si $A = 0$ ou bien si $C = 0$.
- une ellipse si $AC > 0$
- une hyperbole si $AC < 0$

Initialisation

```
> restart;
> with(plots,display,setoptions):
  setoptions(font=["TIMES",roman,10],size=[400,400]):
  with(Student[Precalculus],CompleteSquare):
  with(geometry,conic,detail):
> _EnvHorizontalName := 'x': _EnvVerticalName := 'y':
> libname:="C:\\Users\\plant\\Desktop\\Éléments site Maple au cégep\\Downloads\\Documents Maple\\EED\\Bibliothèque conique",libname:
> with(conique);
```

$[cercleplot, ellipseplotx, ellipseploty, hyperboleplotx, hyperboleploty, paraboleplotx, paraboleploty]$ **(2.1)**

Macro-commandes

cercleplot

cercleplot(Équation,Centre,Rayon,Options)

La macro-commande `Cercleplot()` trace un cercle.

Description des arguments:

– **Équation** est l'équation cartésienne du cercle à tracer

– **Centre** est une liste $[x_0, y_0]$ de deux nombres réels spécifiant le centre du cercle

– **Rayon** est un nombre réel positif spécifiant le rayon

– **Options** est une séquence d'options de la macro-commande plot.

On obtient la superposition des tracés du cercle et celui du point centre avec ses coordonnées.

paraboleplotx et paraboleploty

paraboleplotx(Équation,Sommet,a,Options)

La macro-commande `paraboleplotx()` trace une parabole dont l'axe est parallèle à l'axe des x (parabole à axe horizontal).

Description des arguments:

– **Équation** est l'équation cartésienne de la parabole à tracer

– **Sommet** est une liste $[x_0, y_0]$ de deux nombres réels spécifiant le sommet de la parabole

– ***a*** est un nombre réel spécifiant le coefficient de la variable y^2 dans la forme $x = a y^2 + b y + c$

– **Options** est une séquence d'options de la macro-commande plot.

Lorsque l'axe de la parabole est parallèle à l'axe des y (parabole à axe vertical), la macro-commande à utiliser pour le tracé de la parabole est `paraboleploty()`. Dans ce cas, ***a*** est le coefficient de x^2 dans la forme $y = a x^2 + b x + c$.

paraboleploty(Équation,Centre,a,b,Options)

Dans chaque cas, on obtient la superposition des tracés de la parabole, du foyer avec ses coordonnées et de la directrice. Il y a aussi superposition du tracé de l'axe avec le style tirets.

ellipseplotx et ellipseploty

ellipseplotx(Équation,Centre,a,b,Options)

La macro-commande `ellipseplotx()` trace une ellipse dont l'axe principal (grand axe) est parallèle à l'axe des x.

Description des arguments:

– **Équation** est l'équation cartésienne de l'ellipse à tracer

– **Centre** est une liste $[x_0, y_0]$ de deux nombres réels spécifiant le centre de l'ellipse

– ***a*** est un nombre réel spécifiant la demi longueur de l'axe principal (grand axe)

– ***b*** est un nombre réel spécifiant la demi longueur du second axe (petit axe)

– **Options** est une séquence d'options de la macro-commande plot.

Lorsque l'axe principal est parallèle à l'axe des y, la macro-commande à utiliser pour le tracé de l'ellipse est `ellipseploty()`.

ellipseploty(Équation,Centre,a,b,Options)

Dans chaque cas, on obtient la superposition des tracés de l'ellipse, des points centre et foyers avec leurs coordonnées. Il y a aussi superposition des tracés de l'axe principal et du second axe avec le style tirets ainsi que les tracés des deux directrices.

hyperboleplotx et hyperboleploty

hyperboleplotx(Équation,Centre,a,b,Options)

La macro-commande `hyperboleplotx()` trace les deux branches d'une hyperbole dont l'axe transversal est parallèle à l'axe des x.

Description des arguments:

- **Équation** est l'équation cartésienne de l'hyperbole à tracer
- **Centre** est une liste $[x_0, y_0]$ de deux nombres réels spécifiant le centre de l'hyperbole
- ***a*** est un nombre réel spécifiant la demi longueur de l'axe transversal
- ***b*** est un nombre réel spécifiant la demi longueur de l'axe conjugué
- **Options** est une séquence d'options de la macro-commande plot.

Lorsque l'axe transversal est parallèle à l'axe des y, la macro-commande à utiliser pour le tracé de l'hyperbole est `hyperboleploty()`.

hyperboleploty(Équation,Centre,a,b,Options)

Dans chaque cas, on obtient la superposition des tracés de l'hyperbole, des points centre et foyers avec leurs coordonnées. Il y a aussi superposition des tracés de l'axe transversal, de l'axe conjugué et des deux asymptotes avec le style tirets ainsi que les tracés des deux directrices.

Le cercle

```
> Équation:=x^2+y^2-6*x+4*y+4=0;
```

$$\acute{E}quation := x^2 + y^2 - 6x + 4y + 4 = 0 \qquad \textbf{(4.1)}$$

$A = C$, s'agit-il d'un cercle? Nous allons faire quelques manipulations algébriques pour essayer de le découvrir.

```
> Équation_1:=CompleteSquare(Équation,y);
```

$$\acute{E}quation\_1 := (y+2)^2 + x^2 - 6x = 0 \qquad \textbf{(4.2)}$$

```
> Équation_2:=CompleteSquare(Équation_1,x);
```

$$\acute{E}quation\_2 := (x-3)^2 + (y+2)^2 - 9 = 0 \qquad \textbf{(4.3)}$$

```
> Équation_canonique:=Équation_2 + (9=9);
```

$$\acute{E}quation\_canonique := (x-3)^2 + (y+2)^2 = 9 \qquad \textbf{(4.4)}$$

On constate que **Équation_canonique** est l'équation d'un cercle centré au point (3, -2) et de rayon a = 3. Vérifions si on a bien identifié la nature et les caractéristiques de cette conique.

L'extension geometry est une bibliothèque fort intéressante pour l'étude ou la création d'objets à deux dimensions de la géométrie Euclidienne. Cette bibliothèque permet entre autres d'obtenir l'ensemble des caractéristiques d'une conique quelconque. Chargeons-la afin de vérifier la nature et l'ensemble des caractéristiques qui auront été mises à jour dans l'étude de la conique.

À l'aide de la macro-commande **conic** de la bibliothèque **geometry**, saisisons d'abord l'équation du lieu en lui donnant pour nom C. Il n'est pas nécessaire de saisir l'équation du lieu sous la forme canonique.

```
> conic(C1,Équation,[x,y]):
ellipse: the given equation is indeed a circle
```

Ensuite, à l'aide de la macro-commande **detail** de cette bibliothèque, obtenons directement la nature et les caractéristiques de cette conique.

```
> detail(C1);
```

name of the object	*C1*
form of the object	*circle2d*
name of the center	*center_C1*
coordinates of the center	$[3, -2]$
radius of the circle	$\sqrt{9}$
equation of the circle	$x^2 + y^2 - 6x + 4y + 4 = 0$

(4.5)

La macro-commande de la bibliothèque maison **Cercle()** tracera efficacement la conique identifiée comme un cercle.

```
> Équation:=x^2+y^2-6*x+4*y+4=0;
  Centre:=[3,-2];
  Rayon:=sqrt(9);
```

$$Équation := x^2 + y^2 - 6x + 4y + 4 = 0$$

$$Centre := [3, -2]$$

$$Rayon := 3 \quad \textbf{(4.6)}$$

```
> cercleplot(Équation,Centre,Rayon);
```



Tracez le cercle d'équation $\left(x-\sqrt{3}\right)^2+\left(y+1-\sqrt{2}\right)^2=17$.

```
> cercleplot((x-sqrt(3))^2+(y+1-sqrt(2))^2 = 17,[sqrt(3),-1+sqrt(2)],sqrt(17),
             color="DarkGreen",
             thickness=3,
             linestyle=3,
             filledregions=true,
             coloring = ["Yellow", "Orange"]);
```



La parabole

Cas où l'axe de symétrie est parallèle à l'axe des x

```
> Équation:=y^2-12*y-4*x=-28;
```

$$Équation := y^2-4x-12y=-28 \quad \textbf{(5.1.1)}$$

$C=0$. Devrait-on s'attendre à obtenir une parabole?

```
> Équation_1:=CompleteSquare(Équation,y);
  Équation_canonique:=isolate(Équation_1,x);
```

$$\acute{E}quation\_1 := (y-6)^2 - 4x - 36 = -28$$

$$\acute{E}quation\_canonique := x = -2 + \frac{(y-6)^2}{4} \quad \textbf{(5.1.2)}$$

On constate que **Équation** est l'équation d'une parabole d'axe horizontal ouverte vers la droite ($p > 0$) car $a = \frac{1}{4}$.

Le sommet est le point (-2, 6).

Vérifions si on a bien identifié la nature et les caractéristiques de cette conique.

```
> conic(C2,Équation,[x,y]):
  detail(C2);
```

name of the object	$C2$
form of the object	$parabola2d$
vertex	$[-2, 6]$
focus	$[-1, 6]$
directrix	$x + 3 = 0$
equation of the parabola	$y^2 - 4x - 12y + 28 = 0$

(5.1.3)

Traçons le graphique de cette parabole ouverte vers la droite en se basant directement sur l'ensemble de ses caractéristiques.

```
> Sommet:=[-2,6]:
  a:=1/4:
  paraboleplotx(Équation,Sommet,a,color=orange);
```



Autre exemple.

```
> Équation:=2*x+y^2-8*y=22;
  Équation_1:=CompleteSquare(Équation,y);
  Équation_canonique:=isolate(Équation_1,x);
```

$$\acute{E}quation := y^2 + 2x - 8y = 22$$

$$\acute{E}quation\_1 := (y-4)^2 + 2x - 16 = 22$$

$$\acute{E}quation\_canonique := x = 19 - \frac{(y-4)^2}{2} \quad \textbf{(5.1.4)}$$

On constate que **`Équation`** représente l'équation d'une parabole d'axe horizontal ouverte vers la gauche ($p < 0$) car $a = -\frac{1}{2}$.

Le sommet est le point (19, 4).

```
> conic(C3,Équation,[x,y]):
  detail(C3);
```

name of the object	$C3$	
form of the object	$parabola2d$	
vertex	$[19, 4]$	
focus	$\left[\frac{37}{2}, 4\right]$	**(5.1.5)**
directrix	$x - \frac{39}{2} = 0$	
equation of the parabola	$y^2 + 2x - 8y - 22 = 0$	

```
> Sommet:=[19,4]:
  a:=-1/2:
  paraboleplotx(Équation,Sommet,a,color="Purple",thickness=3);
```



Cas où l'axe de symétrie est parallèle à l'axe des y

```
> Équation:=3*y+4*x^2+8*x=5;
```

$$\acute{E}quation := 4x^2 + 8x + 3y = 5 \quad \textbf{(5.2.1)}$$

$C = 0$. Devrait-on s'attendre à obtenir une parabole?

```
> Équation_1:=CompleteSquare(Équation,x);
```

$$Équation\_1 := 4\,(x+1)^2 + 3\,y - 4 = 5 \quad \textbf{(5.2.2)}$$

```
> Lieu_canonique:=isolate(Équation_1,y);
```

$$Lieu\_canonique := y = 3 - \frac{4\,(x+1)^2}{3} \quad \textbf{(5.2.3)}$$

On constate que **Équation** représente l'équation d'une parabole d'axe vertical ouverte vers le bas avec $a = -\frac{4}{3}$.

Le sommet est le point (-1, 3).

Vérifions si on a bien identifié la nature et les caractéristiques de cette conique.

```
> conic(C4,Équation,[x,y]):
  detail(C4);
```

name of the object	$C4$	
form of the object	$parabola2d$	
vertex	$[-1, 3]$	
focus	$\left[-1, \frac{45}{16}\right]$	**(5.2.4)**
directrix	$y - \frac{51}{16} = 0$	
equation of the parabola	$4\,x^2 + 8\,x + 3\,y - 5 = 0$	

Traçons le graphique de cette parabole d'axe vertical ouverte vers le bas en se basant directement sur l'ensemble de ses caractéristiques.

```
> Sommet:=[-1,3]:
  a:=-4/3:
  paraboleploty(Équation,Sommet,a);
```



Autres exemples:

```
> Équation:=-3*y+4*x^2-8*x=5;
  Équation_1:=CompleteSquare(Équation,x);
  Équation_canonique:=isolate(Équation_1,y);
```

$$Équation := 4x^2 - 8x - 3y = 5$$

$$Équation\_1 := 4(x-1)^2 - 3y - 4 = 5$$

$$Équation\_canonique := y = -3 + \frac{4(x-1)^2}{3} \quad \textbf{(5.2.5)}$$

```
> Sommet:=[1,-3]:
  a:=4/3:
  paraboleploty(Équation,Sommet,a,color=magenta);
```




```
> Équation:=x^2-2*x-12*y=-25;
  Équation_1:=CompleteSquare(Équation,x);
  Équation_canonique:=isolate(Équation_1,y);
```

$$Équation := x^2 - 2x - 12y = -25$$

$$Équation\_1 := (x-1)^2 - 12y - 1 = -25$$

$$Équation\_canonique := y = 2 + \frac{(x-1)^2}{12} \quad \textbf{(5.2.6)}$$

```
> conic(C5,Équation,[x,y]):
  detail(C5);
```

name of the object	$C5$
form of the object	$parabola2d$
vertex	$[1, 2]$
focus	$[1, 5]$
directrix	$y + 1 = 0$
equation of the parabola	$x^2 - 2x - 12y + 25 = 0$

(5.2.7)

```
> Sommet:=[1,2]:
  a:=1/12:
  paraboleploty(Équation,Sommet,a);
```



Tracez la parabole d'équation $y = 12 - (x-6)^2$.

```
> paraboleploty(y = 12-(x-6)^2,[6,12],-1);
```



L'ellipse

Cas où l'axe principal est parallèle à l'axe des x

```
> Équation:=x^2+3*y^2-4*x+6*y=-1;
```

$$\mathit{Équation} := x^2 + 3y^2 - 4x + 6y = -1 \quad \textbf{(6.1.1)}$$

$AC > 0$, s'agit-il d'une ellipse?

```
> Équation_1:=CompleteSquare(Équation,y);
```

$$\mathit{Équation\_1} := 3(y+1)^2 + x^2 - 4x - 3 = -1 \quad \textbf{(6.1.2)}$$

```
> Équation_2:=CompleteSquare(Équation_1,x);
```

$$\acute{E}quation\_2 := (x-2)^2 + 3\,(y+1)^2 - 7 = -1 \tag{6.1.3}$$

```
> Équation_3:=Équation_2 + (7=7);
```

$$\acute{E}quation\_3 := (x-2)^2 + 3\,(y+1)^2 = 6 \tag{6.1.4}$$

```
> Lieu_canonique:=Équation_3 / 6;
```

$$Lieu\_canonique := \frac{(x-2)^2}{6} + \frac{(y+1)^2}{2} = 1 \tag{6.1.5}$$

On constate que **`Équation`** représente donc l'équation d'une ellipse centrée au point (2, -1) avec $a = \sqrt{6}$ et $b = \sqrt{2}$.
Le centre, les sommets et les foyers sont donc situés sur une droite parallèle à l'axe des *x*.

Vérifions si on a bien identifié la nature et les caractéristiques de cette conique.

```
> conic(C6,Équation,[x,y]):
  detail(C6);
```

name of the object	*C6*	
form of the object	*ellipse2d*	
center	$[2, -1]$	
foci	$[[0, -1], [4, -1]]$	**(6.1.6)**
length of the major axis	$2\sqrt{6}$	
length of the minor axis	$2\sqrt{2}$	
equation of the ellipse	$x^2 + 3\,y^2 - 4\,x + 6\,y + 1 = 0$	

```
> Centre:=[2,-1]:
  a:=sqrt(6):
  b:=sqrt(2):
  ellipseplotx(Équation,Centre,a,b,color=orange);
```



Tracez l'ellipse d'équation $\frac{(x+2)^2}{36} + \frac{(y-1)^2}{4} = 1$.

```
> ellipseplotx((x+2)^2/36+(y-1)^2/4=1,[-2,1],6,2,size=[400,300]);
```



Cas où l'axe principal est parallèle à l'axe des *y*

```
> Équation:=4*x^2+y^2-8*x+4*y=-4;
```

$$\text{Équation} := 4x^2 + y^2 - 8x + 4y = -4 \tag{6.2.1}$$

$A\,C > 0$, s'agit-il d'une ellipse?

```
> Équation_1:=CompleteSquare(Équation,y);
```

$$\text{Équation\_1} := (y+2)^2 + 4x^2 - 8x - 4 = -4 \tag{6.2.2}$$

```
> Équation_2:=CompleteSquare(Équation_1,x);
```

$$\text{Équation\_2} := 4(x-1)^2 + (y+2)^2 - 8 = -4 \tag{6.2.3}$$

```
> Équation_3:=Équation_2 + (8=8);
```

$$\text{Équation\_3} := 4(x-1)^2 + (y+2)^2 = 4 \tag{6.2.4}$$

```
> Lieu_canonique:=Équation_3 / 4;
```

$$\text{Lieu\_canonique} := (x-1)^2 + \frac{(y+2)^2}{4} = 1 \tag{6.2.5}$$

On constate que **`Équation`** représente l'équation d'une ellipse centrée au point (1,-2) avec $a = 2$ et $b = 1$. Le centre, les sommets et les foyers sont donc situés sur une droite parallèle à l'axe des *y*.

Vérifions si on a bien identifié la nature et les caractéristiques de cette conique.

```
> conic(C7,Équation,[x,y]):
  detail(C7);
```

(6.2.6)

name of the object	*C7*	
form of the object	*ellipse2d*	
center	$[1, -2]$	
foci	$\left[\left[1, -2-\sqrt{3}\right], \left[1, -2+\sqrt{3}\right]\right]$	**(6.2.6)**
length of the major axis	4	
length of the minor axis	2	
equation of the ellipse	$4x^2 + y^2 - 8x + 4y + 4 = 0$	

```
> Centre:=[1,-2]:
  a:=2:
  b:=1:
  ellipseploty(Équation,Centre,a,b,
               color=orange,
               ytickmarks=8,
               size=[300,400],
               view=[-1..3,-5..1]);
```



Tracez l'ellipse d'équation $\frac{(x-2)^2}{6} + \frac{(y+1)^2}{2} = 1$.

```
> ellipseplotx((x-2)^2/6+(y+1)^2/2 = 1,[2,-1],sqrt(6),sqrt(2));
```



Le cercle comme ellipse dégénérée

La macro-commande usager **Ellipse_x()** ou **Ellipse_y()** tracera efficacement la conique identifiée comme un cercle.

```
> Équation:=-x^2-y^2-4*x+6*y=-1;
```

$$\mathit{Équation} := -x^2 - y^2 - 4x + 6y = -1 \tag{6.3.1}$$

```
> Équation_1:=CompleteSquare(Équation,y);
```

$$\mathit{Équation\_1} := -(y-3)^2 - x^2 - 4x + 9 = -1 \tag{6.3.2}$$

```
> Équation_2:=CompleteSquare(Équation_1,x);
```

$$\mathit{Équation\_2} := -(x+2)^2 - (y-3)^2 + 13 = -1 \tag{6.3.3}$$

```
> Équation_3:=Équation_2 + (-13=-13);
```

$$\mathit{Équation\_3} := -(x+2)^2 - (y-3)^2 = -14 \tag{6.3.4}$$

```
> Équation_canonique:=Équation_3 /(-14);
```

$$\mathit{Équation\_canonique} := \frac{(x+2)^2}{14} + \frac{(y-3)^2}{14} = 1 \tag{6.3.5}$$

```
> Centre:=[-2,3]:
  a:=sqrt(14):
  b:=sqrt(14):
  ellipseploty(Équation,Centre,a,b,color="DarkOrchid",scaling=constrained);
```



Les deux directrices sont repoussées à $\pm\infty$

L'hyperbole

Cas où l'axe transversal est parallèle à l'axe des x

```
> Équation:=2*x^2-3*y^2-4*x-6*y=13;
```

$$\textit{Équation} := 2x^2 - 3y^2 - 4x - 6y = 13 \tag{7.1.1}$$

Cette fois-ci, $A\,C < 0$. Serait-ce une hyperbole?

```
> Équation_1:=CompleteSquare(Équation,x);
```

$$\textit{Équation\_1} := 2(x-1)^2 - 3y^2 - 6y - 2 = 13 \tag{7.1.2}$$

```
> Équation_2:=CompleteSquare(Équation_1,y);
```

$$\textit{Équation\_2} := -3(y+1)^2 + 2(x-1)^2 + 1 = 13 \tag{7.1.3}$$

```
> Équation_3:=Équation_2 + (-1=-1);
```

$$\textit{Équation\_3} := -3(y+1)^2 + 2(x-1)^2 = 12 \tag{7.1.4}$$

```
> Lieu_canonique:=Équation_3 / 12;
```

$$\textit{Lieu\_canonique} := -\frac{(y+1)^2}{4} + \frac{(x-1)^2}{6} = 1 \tag{7.1.5}$$

On constate que **`Équation`** est l'équation d'une hyperbole centrée au point (1, -1) avec $a = \sqrt{6}$ et $b = 2$. Les sommets et les foyers sont donc situés sur la droite (axe transversal) y = -1. L'axe conjugué est la droite x = 1.

Vérifions si on a bien identifié la nature et les caractéristiques de cette conique.

```
> conic(C8,Équation,[x,y]):
  detail(C8);
```

name of the object	*C8*	
form of the object	*hyperbola2d*	
center	$[1, -1]$	
foci	$\left[\left[1-\sqrt{10}, -1\right], \left[1+\sqrt{10}, -1\right]\right]$	
vertices	$\left[\left[1-\sqrt{6}, -1\right], \left[1+\sqrt{6}, -1\right]\right]$	**(7.1.6)**
the asymptotes	$\left[\frac{\sqrt{6}\,x}{3}+y-\frac{\sqrt{6}}{3}+1=0, -\frac{\sqrt{6}\,x}{3}+y+\frac{\sqrt{6}}{3}+1=0\right]$	
equation of the hyperbola	$2x^2-3y^2-4x-6y-13=0$	

La macro-commande usager **Hyperbole_x()** tracera efficacement la conique identifiée comme une hyperbole.

```
> Centre:=[1,-1]:
  a:=sqrt(6):
  b:=2:
  hyperboleplotx(Équation,Centre,a,b,color="DarkRose");
```



Tracez l'hyperbole d'équation $\frac{(x-3)^2}{16}-\frac{(y+1)^2}{36}=1$.

```
> hyperboleplotx((x-3)^2/16-(y+1)^2/36 = 1,[3,-1],4,6);
```



Cas où l'axe transversal est parallèle à l'axe des y

```
> Équation:=-x^2+4*y^2-2*x-16*y=-11;
```

$$\acute{E}quation := -x^2+4y^2-2x-16y=-11 \tag{7.2.1}$$

$A\,C<0$. Sagit-il d'une hyperbole?

```
> Équation_1:=CompleteSquare(Équation,y);
```

$$\acute{E}quation\_1 := 4\,(y-2)^2-x^2-2x-16=-11 \tag{7.2.2}$$

```
> Équation_2:=CompleteSquare(Équation_1,x);
```

$$\acute{E}quation\_2 := -(x+1)^2+4\,(y-2)^2-15=-11 \tag{7.2.3}$$

```
> Équation_3:=Équation_2 + (15=15);
```

$$\acute{E}quation\_3 := -(x+1)^2+4\,(y-2)^2=4 \tag{7.2.4}$$

```
> Équation_canonique:=Équation_3 / 4;
```

$$\acute{E}quation\_canonique := -\frac{(x+1)^2}{4}+(y-2)^2=1 \tag{7.2.5}$$

On constate que **`Équation`** est l'équation d'une hyperbole centrée au point (-1,2) avec $a=1$ et $b=2$. Les sommets et les foyers sont situés sur la droite (axe transverse) $x=-1$. L'axe conjugué est la droite $y=2$.

Vérifions si on a bien identifié la nature et les caractéristiques de cette conique.

```
> conic(C9,Équation,[x,y]):
  detail(C9);
```

(7.2.6)

name of the object	*C9*	
form of the object	*hyperbola2d*	
center	$[-1, 2]$	
foci	$[[-1, 2-\sqrt{5}], [-1, 2+\sqrt{5}]]$	**(7.2.6)**
vertices	$[[-1, 1], [-1, 3]]$	
the asymptotes	$\left[y+\frac{x}{2}-\frac{3}{2}=0, y-\frac{x}{2}-\frac{5}{2}=0\right]$	
equation of the hyperbola	$-x^2+4y^2-2x-16y+11=0$	

La macro-commande usager **Hyperbole_x()** tracera efficacement la conique identifiée comme une hyperbole.

```
> Centre:=[-1,2]:
  a:=1:
  b:=2:
  hyperboleploty(Équation,Centre,a,b);
```



Tracez l'hyperbole équilatère d'équation $\frac{(x+3)^2}{16}-\frac{(y-1)^2}{16}=1$.

```
> conic(C10,(x+3)^2/16-(y-1)^2/16 = 1,[x,y]):
  detail(C10);
```

(7.2.7)

name of the object	*C10*	
form of the object	*hyperbola2d*	
center	$[-3, 1]$	
foci	$[[-3-4\sqrt{2}, 1], [-3+4\sqrt{2}, 1]]$	**(7.2.7)**
vertices	$[[-7, 1], [1, 1]]$	
the asymptotes	$[y+x+2=0, y-x-4=0]$	
equation of the hyperbola	$-\frac{1}{2}+\frac{1}{16}x^2-\frac{1}{16}y^2+\frac{3}{8}x+\frac{1}{8}y=0$	

```
> hyperboleplotx((x+3)^2/16-(y-1)^2/16=1,[-3,1],4,4,color=orange);
```



Les coniques dégénérées

Cas d'un point

```
> Équation:=2*x^2+3*y^2-4*x+24*y+50=0;
```

$$Équation := 2x^2+3y^2-4x+24y+50=0 \quad \textbf{(8.1.1)}$$

$AC > 0$ S'agit-il d'une ellipse?

```
> Équation_1:=CompleteSquare(Équation,y);
```

$$Équation\_1 := 3(y+4)^2+2x^2-4x+2=0 \quad \textbf{(8.1.2)}$$

```
> Équation_2:=CompleteSquare(Équation_1,x);
```

$$Équation\_2 := 2(x-1)^2+3(y+4)^2=0 \quad \textbf{(8.1.3)}$$

On remarque immédistement que c'est la somme de deux termes positifs égalant 0. On doit alors conclure que chacun des termes vaut 0 et donc que **`Équation`** représente l'équation d'un… point, soit le point $(1, -4)$

`Équation` représente donc le point $P(1, -4)$ dans le plan cartésien.
Vérifions si on a bien identifié la nature de cette conique dégénérée.

```
> conic(C10,Équation,[x,y]):
  detail(C10);
```

conic: degenerate case: single point

$$\begin{array}{ll} \text{name of the object} & C10 \\ \text{form of the object} & point2d \\ \text{coordinates of the point} & [1, -4] \end{array} \qquad \textbf{(8.1.4)}$$

Cas de deux droites sécantes

```
> Équation:=3*x^2-4*y^2-6*x-8*y=1;
```

$$\mathit{Équation} := 3x^2 - 4y^2 - 6x - 8y = 1 \qquad \textbf{(8.2.1)}$$

A et C sont de signes contraires. À quoi faut-il donc s'attendre?

```
> Équation_1:=CompleteSquare(Équation,y);
```

$$\mathit{Équation\_1} := -4(y+1)^2 + 3x^2 - 6x + 4 = 1 \qquad \textbf{(8.2.2)}$$

```
> Équation_2:=CompleteSquare(Équation_1,x);
```

$$\mathit{Équation\_2} := 3(x-1)^2 - 4(y+1)^2 + 1 = 1 \qquad \textbf{(8.2.3)}$$

```
> Éq_1:=isolate(Équation_2,(y+1)^2);
```

$$\mathit{Éq\_1} := (y+1)^2 = \frac{3(x-1)^2}{4} \qquad \textbf{(8.2.4)}$$

```
> Éq_2:=map(x->sqrt(x),Éq_1);
```

$$\mathit{Éq\_2} := \sqrt{(y+1)^2} = \frac{\sqrt{3}\sqrt{(x-1)^2}}{2} \qquad \textbf{(8.2.5)}$$

```
> Éq_3:=simplify(Éq_2,assume=real);
```

$$\mathit{Éq\_3} := |y+1| = \frac{\sqrt{3}\,|x-1|}{2} \qquad \textbf{(8.2.6)}$$

```
> D1:=simplify(Éq_3) assuming y >= -1, x >= 1;
  D2:=simplify(Éq_3) assuming y >= -1, x < 1;
```

$$D1 := y+1 = \frac{\sqrt{3}(x-1)}{2}$$

$$D2 := y+1 = -\frac{\sqrt{3}(x-1)}{2} \qquad \textbf{(8.2.7)}$$

Le lieu géométrique **Lieu** est donc deux droites sécantes et ces deux droites se rencontrent clairement au point $(1, -1)$.
Vérifions si on a bien identifié la nature de cette conique dégénérée.

```
> conic(C11,Équation,[x,y]):
  detail(C11);
```

conic: degenerate case: two intersecting lines

name of the object $Line\_1\_C11$
form of the object $line2d$
equation of the line $\frac{\sqrt{3}\,x}{2} + y - \frac{\sqrt{3}}{2} + 1 = 0$,

name of the object $Line\_2\_C11$
form of the object $line2d$
equation of the line $-\frac{\sqrt{3}\,x}{2} + y + \frac{\sqrt{3}}{2} + 1 = 0$ **(8.2.8)**

Traçons cette conique dégénérée.

```
> Droite_1:=plot([x,(1/2)*sqrt(3)*(x-1)-1,x=-3..5],color=navy):
  Droite_2:=plot([x,-(1/2)*sqrt(3)*(x-1)-1,x=-3..5],color=orange):
> display({Droite_1,Droite_2},scaling=constrained);
```



Cas d'un lieu imaginaire

```
> Équation:=2*x^2+5*y^2-4*x+10*y=-8;
```

$$Équation := 2x^2 + 5y^2 - 4x + 10y = -8 \quad \textbf{(8.3.1)}$$

A et C sont de même signe. Alors...

```
> Équation_1:=CompleteSquare(Équation,y);
```

$$Équation\_1 := 5(y+1)^2 + 2x^2 - 4x - 5 = -8 \quad \textbf{(8.3.2)}$$

```
> Équation_2:=CompleteSquare(Équation_1,x);
```

$$Équation\_2 := 2(x-1)^2 + 5(y+1)^2 - 7 = -8 \quad \textbf{(8.3.3)}$$

```
> Équation_3:=Équation_2+(7=7);
```

$$Équation\_3 := 2(x-1)^2 + 5(y+1)^2 = -1 \quad \textbf{(8.3.4)}$$

La somme de deux termes positifs égalerait un nombre négatif !!!
L'équation admet évidemment aucune racine.
Il n'y a aucne représentation graphique de ce lieu imaginaire.
Vérifions si on a bien identifié la nature de cette conique dégénérée.

```
> conic(C12,Équation,[x,y]):
  detail(C12);
conic: there is no locus
Error, (in geometry:-detail) unknown object:  C12
```

Exercices

Indentifiez la nature de la conique ou de la conique dégénérée représentée par chaque équation.
Tracez, si possible, le lieu de chaque équation.

a) $x^2 + y^2 - 2x + 2y + 1 = 0$
b) $x^2 + y^2 + 6x - 2y + 6 = 0$
c) $9x^2 + 4y^2 + 72x - 16y + 124 = 0$
d) $16x^2 + 9y^2 + 192x + 90y + 1000 = 0$
e) $y^2 - 5x - 4y - 6 = 0$
f) $3x^2 + 3y^2 - 6x + 12y + 60 = 0$
g) $4x^2 - 4y^2 + 8x - 12y - 5 = 0$
h) $4x^2 - 4y^2 + 8x - 12y - 6 = 0$
i) $25x^2 + 4y^2 + 150x - 8y + 129 = 0$